

```

600 READ K$:IF S>0 THEN 640
610 FOR L=1 TO LEN(I$)-LEN(K$)+1
620 IF MID$(I$,L,LEN(K$))=K$ THEN S=K:T=L:F$=K$
630 NEXT L
640 NEXT K
650 IF S>0 THEN K=S:L=T:GOTO 680
660 K=36:GOTO 910: REM NO KEYWORDS
670 REM
680 REM ♦♦ TAKE RIGHT PART OF STRING AND
690 REM     CONJUGATE IT, USING LIST OF STRINGS
700 REM     TO BE SWAPPED.
710 REM
720 RESTORE 1410
730 C$=" "+RIGHT$(I$,LEN(I$)-LEN(F$)-L+1)+" "
740 FOR X=1 TO N2/2:READ S$,R$
750 FOR L=1 TO LEN(C$)
760 IF L+LEN(S$)>LEN(C$) THEN 800
770 IF MID$(C$,L,LEN(S$))<>S$ THEN 800
780 C$=LEFT$(C$,L-1)+R$+RIGHT$(C$,LEN(C$)-L-LEN(S$)+1)
790 L=L+LEN(R$):GOTO 830
800 IF L+LEN(R$)>LEN(C$) THEN 830
810 IF MID$(C$,L,LEN(R$))<>R$ THEN 830
820 C$=LEFT$(C$,L-1)+S$+RIGHT$(C$,LEN(C$)-L-LEN(R$)+1)
830 NEXT L
840 NEXT X
850 IF MID$(C$,2,1)=" " THEN C$=RIGHT$(C$,LEN(C$)-1)
860 FOR L=1 TO LEN(C$)
870 IF MID$(C$,L,1)<>"!" THEN 890
880 C$=LEFT$(C$,L-1)+RIGHT$(C$,LEN(C$)-L):GOTO 870
890 NEXT L
900 REM
910 REM ♦♦ NOW USING KEYWORD NUMBER (K),
920 REM     GET THE REPLY.
930 REM
940 RESTORE 1480
950 FOR X=1 TO R(K):READ F$:NEXT
960 R(K)=R(K)+1:IF R(K)>N(K) THEN R(K)=S(K)
970 REM
980 REM ♦♦ CONVERT OUTPUT STRINGS TO LOWER CASE.
990 REM
1000 IF RIGHT$(F$,1)<>"♦" THEN Z$=F$:GOTO 1020
1010 Z$=LEFT$(F$,LEN(F$)-1)+C$
1020 Z1$="":FOR L=1 TO LEN(Z$)
1030 Z=ASC(MID$(Z$,L,1)):IF L=1 THEN 1070
1040 IF MID$(Z$,L,3)<>" I " THEN 1060
1050 Z1$=Z1$+" I ":L=L+2:GOTO 1030
1060 IF Z>=65 THEN Z=Z+32
1070 Z1$=Z1$+CHR$(Z)
1080 NEXT
1090 Z2$=RIGHT$(Z$,1)
1100 IF Z2$="."OR Z2$="?"OR Z2$="!" THEN 1170
1110 FOR L=LEN(Z1$) TO 1 STEP -1
1120 IF MID$(Z1$,L,1)=" " THEN NEXT
1130 Z1$=LEFT$(Z1$,L)+" ?"
1140 REM
1150 REM ♦♦ JUSTIFY TO WIDTH HELD IN A
1160 REM
1170 IF LEN(Z1$)<A THEN PRINT Z1$:GOTO 1230
1180 Z$=LEFT$(Z1$,A):FOR J=LEN(Z$) TO 1 STEP -1
1190 IF MID$(Z$,J,1)=" " THEN 1210
1200 NEXT
1210 Z$=LEFT$(Z$,J):PRINT Z$
1220 Z1$=RIGHT$(Z1$,LEN(Z1$)-J):GOTO 1170
1230 P$=I$:GOTO 310

```